

```

stop();

//a next connection object is required to connect to streaming movies
//it deals primarily with the flash media server
var nc:NetConnection = new NetConnection();
//since we are not actually dealing with a media server, connect to nothing
nc.connect(null);

//now create a net stream based off our connection
var ns:NetStream = new NetStream(nc);
//an event in case we get an error with our synching
ns.addEventListener(AsyncErrorEvent.ASYNC_ERROR, asyncErrorHandler);
function asyncErrorHandler(event:AsyncErrorEvent):void
{
    // ignore error
    // we go though this trouble so flash does not display a popup error to the user
}

//finally we specify the URL of the movie we wish to play.
ns.play("video.flv");

//next we create a video object. This is the piece that
// is actually 'visible' to the user
var vid:Video = new Video();
// and we 'skin' it with the video we're getting from the net
vid.attachNetStream(ns);
//position it
vid.x = 100;
vid.y = 50;
//name it for future reference
vid.name = "loadedMovie";
//add it to the stage
addChild(vid);

//listeners for button presses
pauseBtn.addEventListener(MouseEvent.CLICK, pauseHandler);
playBtn.addEventListener(MouseEvent.CLICK, playHandler);
stopBtn.addEventListener(MouseEvent.CLICK, stopHandler);

function pauseHandler(event:MouseEvent):void
{
    //pause the net stream.
    ns.pause();
}
function playHandler(event:MouseEvent):void
{
    //resume the net stream
    ns.resume();
}
function stopHandler(event:MouseEvent):void
{
    // Pause the stream and move the playhead back to
    // the beginning of the stream.
    ns.pause();
    ns.seek(0);
}

```